


Introduction to Programming in SAS Viya

Stacey Syphus

Director, SAS Education

Slides





SAS Compute Server

```
libname pvbase "&path/data";  
  
data profit;  
  set pvbase.orders;  
  ...  
run;  
  
proc means data=profit;  
  ...  
run;
```

Continent Name	N Obs	Sum	Mean
Africa	770	-127.4802000	-0.1655182
Asia	1110	15503.70	13.9673063
Europe	653484	5459450.59	8.3558321
North America	235708	2121645.57	9.0014211
Oceania	60397	462934.63	7.6648811

Copyright © SAS Institute Inc. All rights reserved.



```

proc casutil;
  load <options>;
  contents <options>;
quit;

data casuser.profit;
  set casuser.orders;
  ...
run;

proc mdsummary data=casuser.profit;
  var RetailPrice;
  output out=mycas.orders_sum;
run;

```

SAS Cloud Analytic Services (CAS)

CAS-enabled steps share similar syntax with traditional SAS code but execute in CAS.

Copyright © SAS Institute Inc. All rights reserved.

sas

```

proc casutil;
  load <options>;
  contents <options>;
quit;

data casuser.profit;
  set casuser.orders;
  ...
run;

proc mdsample data=casuser.profit;
  var RetailPrice;
  output out=mycas.orders_sum;
run;

```

SAS Cloud Analytic Services (CAS)

CAS-enabled steps share similar syntax with traditional SAS code but execute in CAS.



SAS Viya, CAS-enabled steps

Configure Environment

- Start CAS session

Access Data

- CASLIB statement
- PROC CASUTIL, LOAD statement

Explore Data

- PROC CASUTIL, CONTENTS statement
- DATA step

Generate Reports

- PROC FREQTAB
- PROC MDSUMMARY



SAS Cloud Analytic Services (CAS)

CAS language (CASL)

You can use the CAS procedure and CASL to call CAS actions.


```

proc cas;
  table.loadTable ...;
  dataStep.runCode ...;
  simple.summary...;
quit;

```




- An action set is a collection of CAS actions.
- A CAS action performs a single task.
- Action are defined as *action-set.action*.



Copyright © SAS Institute Inc. All rights reserved.




CASL and CAS actions execute only in CAS.



Copyright © SAS Institute Inc. All rights reserved.



SAS Viya, CAS Language (CASL)



Copyright © SAS Institute Inc. All rights reserved.



Is CAS always faster?

It depends ...

Compute Server

CAS

Copyright © SAS Institute Inc. All rights reserved.

SAS

Why isn't CAS always faster?

- Start services
- Load data into memory
- Distribute data and processing among worker nodes
- Return results to controller
- Deliver combined results to the client

Copyright © SAS Institute Inc. All rights reserved.

SAS

Which steps should I modify to run in CAS?

- steps using data sources larger than 50 GB
- long-running steps (30 minutes+)
- computationally demanding PROCs
- DATA steps with many computations, functions, or conditional logic

Copyright © SAS Institute Inc. All rights reserved.

SAS

Demonstration Programs

Compute Server

```
title "Compute Server Program";
libname pvbase "/home/student/Courses/PGVY01/data";

proc contents data=pvbase.bigOrders;
run;
title;

data pvbase.bigOrders_clean;
  set pvbase.bigOrders;
  Name=catx(' ',
            scan(Customer_Name,2,' '),
            scan(Customer_Name,1,' '));
run;

proc freq data=pvbase.bigOrders;
  tables Country OrderType;
run;

proc means data=pvbase.bigOrders;
  var RetailPrice;
  output out=Orders_sum;
run;
```

CAS Server with CAS-Enabled Steps

```
title "CAS-Enabled Steps";
cas mySession;
caslib mycas path="/home/student/Courses/PGVY01/data" libref=mycas;

proc casutil;
  load casdata="bigOrders.sashdat" incaslib="mycas"
      outcaslib="mycas" casout="bigOrders" replace;
  contents casdata="bigOrders" incaslib="mycas";
quit;
title;

data mycas.bigOrders_clean;
  set mycas.bigOrders;
  Name=catx(' ',
            scan(Customer_Name,2,' '),
            scan(Customer_Name,1,' '));
run;

proc freqtab data=mycas.bigOrders_clean;
  table Country OrderType;
run;
```

```

proc mdsummary data=mycas.bigOrders_clean;
  var RetailPrice;
  output out=mycas.orders_sum;
run;

proc print data=mycas.orders_sum;
run;

cas mySession terminate;

```

CAS Server with CASL

```

cas mySession;

proc cas;
  table.addCaslib /
    name="mycas",
    path="/home/student/Courses/PGVY01/data";

  table.loadTable /
    path="bigOrders.sashdat", caslib="mycas",
    casOut={name="bigOrders", caslib="mycas", replace=true};

  table.columnInfo /
    table={name="bigOrders", caslib="mycas"};

  dataStep.runCode /
    code=
    "
    data mycas.bigOrders_clean;
      set mycas.bigOrders;
      Name=catx(' ',
                scan(Customer_Name,2,' '),
                scan(Customer_Name,1,' '));

    run;
    ";

  simple.summary /
    table={name="bigOrders", caslib="mycas"},
    input={"RetailPrice"},
    casOut={name='orders_sum', replace=true};

  table.fetch /
    table={name="orders_sum", caslib="mycas"};
quit;
title;

cas mySession terminate;

```

Questions and Answers

- 1) **What free resources are available to learn more about programming in SAS Viya?**
 - a. We have a collection of free videos you can use to get started:
<https://video.sas.com/category/videos/an-introduction-to-sas-viya-programming-for-sas-9-programmers>
 - b. Check out these blog posts by Peter Styliadis. He writes about CASL and Python integration: <https://blogs.sas.com/content/author/peterstyliadis/>
 - c. We also have complete courses that cover these topics in depth. You can try the [SAS Learning Subscription](#) that is free for 30 days. The *Programming for SAS Viya* and *High-Performance Data Processing with CASL in SAS Viya* are very thorough. Both courses include access to practice software so you can try it out.

- 2) **Is SAS continuing to add features/enhancements to the SAS Compute Server, or should we think of the compute server as a 'legacy environment'? Meaning, what is the future for non-CAS SAS?**

The Compute Server is a valuable, mature and well-optimized platform for processing reasonably sized data. The SAS Compute Server will continue to be a mainstay of SAS Viya for the foreseeable future. Towards the end of this presentation, Stacey describes the relationship between CAS and Compute Server and give rules of thumb for choosing the best server to run your process.

- 3) **What do we do if we want to save data to our hard disc instead of CAS?**

PROC CASUTIL includes a SAVE statement to allow you to save in-memory tables to permanent storage.

- 4) **Is it a must for one to use double quotation?**

In this code (PROC CASUTIL and PROC CAS), either double or single quotes are acceptable.

- 5) **For those of us who only analyze reasonably sized data, should we expect continued enhancements to the compute server over time? Is there an option of licensing only a compute server without licensing CAS?**

Enhancements to the Compute Server are certainly under development and released using the same mechanism as other Viya updates. SAS has released a [containerized version of SAS Analytics Pro](#) that provides the Viya Compute Server capabilities of Base SAS, SAS/Graph, SAS/STAT and SAS/Access for those that don't need immediate access to CAS.

- 6) **Is there a learning environment for Viya/CAS, similar to SAS On-demand for Academics?**

If you are an educator or student, you can sign up to access [Viya for Learners](#) with a university email. If you are not eligible to use Viya for Learners, access to practice software is included with most of our SAS Viya training courses. You can take advantage of the free 30 day trial to our [SAS Learning Subscription](#), which includes most of our SAS Education e-learning portfolio.

- 7) **Any cheat sheet available to differentiate normal SAS code vs. CAS code?**

Here is a link to the CAS-enabled procedures list:

https://go.documentation.sas.com/doc/en/pgmsascdc/9.4_3.5/proc/p0nnkdmqmqz48w8n1kqof

[zc7mcla4.htm](#)

8) **Do we have a complete documentation on SAS Viya?**

That's like asking "Do we have complete documentation for SAS". The answer is yes, of course, but that's a BIG body of documentation! For SAS Viya programmers, here is a good start:

https://go.documentation.sas.com/doc/en/pgmsascdc/v_017/pgmsaswlcm/home.htm

9) **Can you use PROC SQL code in SAS Viya? Is there a CAS language equivalent?**

PROC SQL runs only on the Compute Server. To do similar SQL processing in CAS, you would use PROC FEDSQL.